# NAG Library Routine Document

# E04UQF/E04UQA

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

To supply optional parameters to E04USF/E04USA from an external file. More precisely, E04UQF must be used to supply optional parameters to E04USF and E04UQA must be used to supply optional parameters to E04USA.

E04UQA is a version of E04UQF that has additional parameters in order to make it safe for use in multithreaded applications (see Section 5). The initialization routine E04WBF **must** have been called before calling E04UQA.

## 2 Specification

### 2.1 Specification for E04UQF

```
SUBROUTINE E04UQF(IOPTNS, INFORM)
INTEGER           IOPTNS, INFORM
```

### 2.2 Specification for E04UQA

```
SUBROUTINE E04UQA(IOPTNS, LWSAV, IWSAV, RWSAV, INFORM)
INTEGER           IOPTNS, IWSAV(610), INFORM
double precision  RWSAV(475)
LOGICAL           LWSAV(120)
```

## 3 Description

E04UQF/E04UQA may be used to supply values for optional parameters to the corresponding routines E04USF/E04USA. E04UQF/E04UQA reads an external file and each line of the file defines a single optional parameter. It is only necessary to supply values for those parameters whose values are to be different from their default values.

Each optional parameter is defined by a single character string, of up to 72 characters, consisting of one or more items. The items associated with a given option must be separated by spaces, or equals signs [=]. Alphabetic characters may be upper or lower case. The string

```
Print Level = 1
```

is an example of a string used to set an optional parameter. For each option the string contains one or more of the following items:

– a mandatory keyword;

– a phrase that qualifies the keyword;

– a number that specifies an INTEGER or ***double precision*** value. Such numbers may be up to 16 contiguous characters in Fortran's I, F, E or D formats, terminated by a space if this is not the last item on the line.

Blank strings and comments are ignored. A comment begins with an asterisk (*) and all subsequent characters in the string are regarded as part of the comment.

The file containing the options must start with `Begin` and must finish with `End`. An example of a valid options file is:

```
Begin * Example options file
   Print level = 5
End
```

For E04UQF each line of the file is normally printed as it is read, on the current advisory message unit (see X04ABF), but printing may be suppressed using the keyword Nolist. To suppress printing of `Begin`, Nolist must be the first option supplied as in the file:

```
Begin
   Nolist
   Print level = 5
End
```

Printing will automatically be turned on again after a call to E04USF/E04USA or E04UQF and may be turned on again at any time using the keyword List.

For E04UQA printing is turned off by default, but may be turned on at any time using the keyword List.

Optional parameter settings are preserved following a call to E04USF/E04USA and so the keyword Defaults is provided to allow you to reset all the optional parameters to their default values before a subsequent call to E04USF/E04USA.

A complete list of optional parameters, their abbreviations, synonyms and default values is given in Section 11 in E04USF/E04USA.

## 4 References

None.

## 5 Parameters

1: IOPTNS – INTEGER *Input*

*On entry*: the unit number of the options file to be read.

*Constraint*: $0 \leq \text{IOPTNS} \leq 99$.

2: INFORM – INTEGER *Output*

**Note**: *for E04UQA, INFORM does not occur in this position in the parameter list. See the additional parameters described below.*

*On exit*: contains zero if the options file has been successfully read and a value $> 0$ otherwise (see Section 6).

**Note**: *the following are additional parameters for specific use with E04UQA. Users of E04UQF therefore need not read the remainder of this description.*

2: LWSAV(120) – LOGICAL array *Communication Array*

3: IWSAV(610) – INTEGER array *Communication Array*

4: RWSAV(475) – ***double precision*** array *Communication Array*

The arrays LWSAV, IWSAV and RWSAV **must not** be altered between calls to any of the routines E04UQA, E04URA, E04USA or E04WBF.

5: INFORM – INTEGER *Output*

**Note**: see the parameter description for INFORM above.

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFORM = 1

IOPTNS is not in the range $[0, 99]$.

INFORM = 2

`Begin` was found, but end-of-file was found before `End` was found.

INFORM = 3

end-of-file was found before `Begin` was found.

INFORM = 4

Not used.

INFORM = 5

One or more lines of the options file is invalid. Check that all keywords are neither ambiguous nor misspelt.

## 7 Accuracy

Not applicable.

## 8 Further Comments

E04URF/E04URA may also be used to supply optional parameters to E04USF/E04USA.

## 9 Example

This example solves the same problem as the example for E04USF/E04USA, but in addition illustrates the use of E04UQF/E04UQA and E04URF/E04URA to set optional parameters for E04USF/E04USA.

In this example the options file read by E04UQF/E04UQA is appended to the data file for the program (see Section 9.2). It would usually be more convenient in practice to keep the data file and the options file separate.

### 9.1 Program Text

**Note**: *the following program illustrates the use of E04UQF. An equivalent program illustrating the use of E04UQA is available with the supplied Library and is also available from the NAG web site.*

```
*     E04UQF Example Program Text
*     Mark 20 Revised. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NMAX, NCLMAX, NCNMAX
      PARAMETER        (MMAX=50,NMAX=10,NCLMAX=10,NCNMAX=10)
      INTEGER          LDA, LDCJ, LDFJ, LDR
      PARAMETER        (LDA=NCLMAX,LDCJ=NCNMAX,LDFJ=MMAX,LDR=NMAX)
      INTEGER          LIWORK, LWORK
      PARAMETER        (LIWORK=100,LWORK=1000)
*     .. Local Scalars ..
      DOUBLE PRECISION OBJF
      INTEGER          I, IFAIL, INFORM, ITER, J, M, N, NCLIN, NCNLN,
     +                 OUTCHN
      LOGICAL          LMOK
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), BL(NMAX+NCLMAX+NCNMAX),
```

```
     +                 BU(NMAX+NCLMAX+NCNMAX), C(NCNMAX),
     +                 CJAC(LDCJ,NMAX), CLAMDA(NMAX+NCLMAX+NCNMAX),
     +                 F(MMAX), FJAC(LDFJ,NMAX), R(LDR,NMAX), USER(1),
     +                 WORK(LWORK), X(NMAX), Y(MMAX)
      INTEGER          ISTATE(NMAX+NCLMAX+NCNMAX), IUSER(1),
     +                 IWORK(LIWORK)
*     .. External Functions ..
      LOGICAL          A00ACF
      EXTERNAL         A00ACF
*     .. External Subroutines ..
      EXTERNAL         CONFUN, E04UQF, E04URF, E04USF, OBJFUN, X04ABF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E04UQF Example Program Results'
      OUTCHN = NOUT
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      READ (NIN,*) NCLIN, NCNLN
      LMOK = A00ACF()
      IF ( .NOT. LMOK) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*) ' ** A valid licence key was not found'
      ELSE IF (M.GT.MMAX .OR. N.GT.NMAX .OR. NCLIN.GT.NCLMAX .OR.
     +         NCNLN.GT.NCNMAX) THEN
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     ' ** At least one of M, N, NCLIN or NCNLN is too large'
      ELSE
*
*        Read A, Y, BL, BU and X from data file
*
         IF (NCLIN.GT.0) READ (NIN,*) ((A(I,J),J=1,N),I=1,NCLIN)
         READ (NIN,*) (Y(I),I=1,M)
         READ (NIN,*) (BL(I),I=1,N+NCLIN+NCNLN)
         READ (NIN,*) (BU(I),I=1,N+NCLIN+NCNLN)
         READ (NIN,*) (X(I),I=1,N)
*
*        Set three options using E04URF
*
         CALL E04URF(' Infinite Bound Size = 1.0D+25 ')
*
         CALL E04URF(' Print Level = 1 ')
*
         CALL E04URF(' Verify Level = -1 ')
*
*        Set the unit number for advisory messages to OUTCHN
*
         CALL X04ABF(1,OUTCHN)
*
*        Read the options file for the remaining options
*
         CALL E04UQF(NIN,INFORM)
*
         IF (INFORM.NE.0) THEN
            WRITE (NOUT,99999) 'E04UQF terminated with INFORM = ',
     +        INFORM
            GO TO 20
         END IF
*
*        Solve the problem
*
         IFAIL = 1
*
         CALL E04USF(M,N,NCLIN,NCNLN,LDA,LDCJ,LDFJ,LDR,A,BL,BU,Y,CONFUN,
     +               OBJFUN,ITER,ISTATE,C,CJAC,F,FJAC,CLAMDA,OBJF,R,X,
     +               IWORK,LIWORK,WORK,LWORK,IUSER,USER,IFAIL)
*
         IF (IFAIL.NE.0) THEN
            WRITE (NOUT,99999) IFAIL
         END IF
      END IF
```

```
   20 CONTINUE
*
99999 FORMAT (1X,/' ** E04USF returned with IFAIL = ',I5)
      END
      SUBROUTINE OBJFUN(MODE,M,N,LDFJ,NEEDFI,X,F,FJAC,NSTATE,IUSER,USER)
*     Routine to evaluate the subfunctions and their 1st derivatives.
*     .. Parameters ..
      DOUBLE PRECISION PT49, ONE, EIGHT
      PARAMETER        (PT49=0.49D0,ONE=1.0D0,EIGHT=8.0D0)
*     .. Scalar Arguments ..
      INTEGER          LDFJ, M, MODE, N, NEEDFI, NSTATE
*     .. Array Arguments ..
      DOUBLE PRECISION F(*), FJAC(LDFJ,*), USER(*), X(N)
      INTEGER          IUSER(*)
*     .. Local Scalars ..
      DOUBLE PRECISION AI, TEMP, X1, X2
      INTEGER          I
      LOGICAL          MODE02, MODE12
*     .. Local Arrays ..
      DOUBLE PRECISION A(44)
*     .. Intrinsic Functions ..
      INTRINSIC        EXP
*     .. Data statements ..
      DATA             A/8.0D0, 8.0D0, 10.0D0, 10.0D0, 10.0D0, 10.0D0,
     +                 12.0D0, 12.0D0, 12.0D0, 12.0D0, 14.0D0, 14.0D0,
     +                 14.0D0, 16.0D0, 16.0D0, 16.0D0, 18.0D0, 18.0D0,
     +                 20.0D0, 20.0D0, 20.0D0, 22.0D0, 22.0D0, 22.0D0,
     +                 24.0D0, 24.0D0, 24.0D0, 26.0D0, 26.0D0, 26.0D0,
     +                 28.0D0, 28.0D0, 30.0D0, 30.0D0, 30.0D0, 32.0D0,
     +                 32.0D0, 34.0D0, 36.0D0, 36.0D0, 38.0D0, 38.0D0,
     +                 40.0D0, 42.0D0/
*     .. Executable Statements ..
      X1 = X(1)
      X2 = X(2)
      MODE02 = MODE .EQ. 0 .OR. MODE .EQ. 2
      MODE12 = MODE .EQ. 1 .OR. MODE .EQ. 2
      DO 20 I = 1, M
         IF (NEEDFI.EQ.I) THEN
            F(I) = X1 + (PT49-X1)*EXP(-X2*(A(I)-EIGHT))
            RETURN
         ELSE
            AI = A(I)
            TEMP = EXP(-X2*(AI-EIGHT))
            IF (MODE02) F(I) = X1 + (PT49-X1)*TEMP
            IF (MODE12) THEN
               FJAC(I,1) = ONE - TEMP
               FJAC(I,2) = -(PT49-X1)*(AI-EIGHT)*TEMP
            END IF
         END IF
   20 CONTINUE
*
      RETURN
      END
*
      SUBROUTINE CONFUN(MODE,NCNLN,N,LDCJ,NEEDC,X,C,CJAC,NSTATE,IUSER,
     +                  USER)
*     Routine to evaluate the nonlinear constraint and its 1st
*     derivatives.
*     .. Parameters ..
      DOUBLE PRECISION ZERO, PT09, PT49
      PARAMETER        (ZERO=0.0D0,PT09=0.09D0,PT49=0.49D0)
*     .. Scalar Arguments ..
      INTEGER          LDCJ, MODE, N, NCNLN, NSTATE
*     .. Array Arguments ..
      DOUBLE PRECISION C(*), CJAC(LDCJ,*), USER(*), X(N)
      INTEGER          IUSER(*), NEEDC(*)
*     .. Local Scalars ..
      INTEGER          I, J
*     .. Executable Statements ..
      IF (NSTATE.EQ.1) THEN
*        First call to CONFUN.  Set all Jacobian elements to zero.
```

```
*        Note that this will only work when 'Derivative Level = 3'
*        (the default; see Section 11.2).
         DO 40 J = 1, N
            DO 20 I = 1, NCNLN
               CJAC(I,J) = ZERO
   20       CONTINUE
   40    CONTINUE
      END IF
*
      IF (NEEDC(1).GT.0) THEN
         IF (MODE.EQ.0 .OR. MODE.EQ.2) C(1) = -PT09 - X(1)*X(2) +
     +       PT49*X(2)
         IF (MODE.EQ.1 .OR. MODE.EQ.2) THEN
            CJAC(1,1) = -X(2)
            CJAC(1,2) = -X(1) + PT49
         END IF
      END IF
*
      RETURN
      END
```

## 9.2 Program Data

```
E04UQF Example Program Data
  44  2                                :Values of M and N
  1   1                                :Values of NCLIN and NCNLN
  1.0   1.0                            :End of matrix A
  0.49  0.49  0.48  0.47  0.48  0.47  0.46  0.46  0.45  0.43  0.45
  0.43  0.43  0.44  0.43  0.43  0.46  0.45  0.42  0.42  0.43  0.41
  0.41  0.40  0.42  0.40  0.40  0.41  0.40  0.41  0.41  0.40  0.40
  0.40  0.38  0.41  0.40  0.40  0.41  0.38  0.40  0.40  0.39  0.39 :End of Y
  0.4      -4.0      1.0      0.0      :End of BL
  1.0E+25   1.0E+25  1.0E+25  1.0E+25  :End of BU
  0.4   0.0                            :End of X
Begin   Example options file for  E04UQF
   Major Iteration Limit   =  15      *  (Default =  50)
   Minor Iteration Limit   =  10      *  (Default =  50)
End
```

## 9.3 Program Results

```
 E04UQF Example Program Results


 Calls to E04URF
 ---------------

       Infinite Bound Size = 1.0D+25
       Print Level = 1
       Verify Level = -1


 OPTIONS file
 ------------

      Begin   Example options file for  E04UQF
         Major Iteration Limit   =  15      *  (Default =  50)
         Minor Iteration Limit   =  10      *  (Default =  50)
      End

 *** E04USF

 Parameters
 ----------

 Linear constraints.....         1       Variables.............          2
 Nonlinear constraints..         1       Subfunctions...........        44

 Infinite bound size....  1.00E+25       COLD start.............
 Infinite step size.....  1.00E+25       EPS (machine precision)  1.11E-16
```

```
 Step limit.............  2.00E+00      Hessian.................       NO

 Linear feasibility.....  1.05E-08      Crash tolerance........  1.00E-02
 Nonlinear feasibility..  1.05E-08      Optimality tolerance...  3.26E-12
 Line search tolerance..  9.00E-01      Function precision.....  4.37E-15

 Derivative level.......         3      Monitoring file........       -1
 Verify level...........        -1

 Major iterations limit.        15      Major print level......        1
 Minor iterations limit.        10      Minor print level......        0

 J'J initial Hessian....                Reset frequency........        2

 Workspace provided is     IWORK(     100),  WORK(    1000).
 To solve problem we need  IWORK(       9),  WORK(     306).

 Exit from NP problem after     6 major iterations,
                                8 minor iterations.


 Varbl State     Value       Lower Bound   Upper Bound    Lagr Mult   Slack

 V   1    FR   0.419953      0.400000          None           .      1.9953E-02
 V   2    FR    1.28485      -4.00000          None           .       5.285


 L Con State     Value       Lower Bound   Upper Bound    Lagr Mult   Slack

 L   1    FR    1.70480       1.00000          None           .      0.7048


 N Con State     Value       Lower Bound   Upper Bound    Lagr Mult   Slack

 N   1    LL  -9.768159E-13       .            None      3.3358E-02 -9.7682E-13

 Exit E04USF - Optimal solution found.

 Final objective value =   0.1422983E-01
```